

```

{:Scene ("/tut/tut2.mhg" 0)
:Items (

// Resident programs ---
// -----
{:ResidentPrg prgCastToObjectRef :InitiallyActive false :Name 'CTO' }

// Constants -----
// -----
{:IntegerConst constTimerDelay :ConstValue 80} // 12.5 FPS
{:IntegerConst constNumBitmaps :ConstValue 5}
{:IntegerConst constNumFrames :ConstValue [constNumBitmaps * 2 - 2]}
{:IntegerConst constPacObjectIdOffset :ConstValue 100}

// Variables -----
// -----
{:BooleanVar boolTemp :OrigValue false}
{:IntegerVar intCount :OrigValue 0}
{:IntegerVar intNextPacObjectId :OrigValue 0}
{:ObjectRefVar objPac :OrigValue :ObjectRef bmpPac0}

// Ingredients -----
// -----
foreach num (0..4) {
  {:Bitmap bmpPac<num> = [constPacObjectIdOffset + <num>]
  :OrigPosition [(720 - 32) / 2] [(576 - 32) / 2]
  :OrigBoxSize 32 32
  :OrigContent :ContentRef ('/tut/<num>.png')
  :InitiallyActive false
}
} endfor

{:Timer tmrTimer}

// Sequentials -----
// -----
defSequential calculateNextFrame :Namespace tgPac {
(
  // increment counter
  intCount.Add(1)
  intCount.Modulo(constNumFrames)

  // calculate object Id of Pac Bitmap to display this frame
  intNextPacObjectId.SetVariable(:IndirectRef intCount)
  :If (intNextPacObjectId >= constNumBitmaps) {
    intNextPacObjectId.SetVariable(constNumFrames)
    intNextPacObjectId.Subtract(:IndirectRef intCount)
  }
  intNextPacObjectId.Add(constPacObjectIdOffset)
)
} endSequential

defSequential updateScreen :Namespace tgPac {
(
  // stop the bitmap of previous frame
  :Stop(:IndirectRef objPac)
  prgCastToObjectRef.Call(boolTemp
    "/tut/tut2.mhg" :IndirectRef intNextPacObjectId
    :IndirectRef objPac)
  // run the bitmap of this frame
  :Run(:IndirectRef objPac)
)
} endSequential

{:TokenGroup tgPac
:TokenGroupItems ((bmpPac0
:ActionSlots (
  insertSequentials (:NameSpace tgPac)
)
))
}

// Links -----
// -----

```

```

{:Link lnkStartup
 :EventSource      0
 :EventType        IsRunning
 :LinkEffect
 (
   // preload graphics
   foreach num (0..4) {
     bmpPac<num>.Preload()
   } endfor

   :SetTimer(0 tmrTimer constTimerDelay)
 )
}

{:Link lnkGameTick
 :EventSource      0
 :EventType        TimerFired
 :EventData        tmrTimer
 :LinkEffect
 (
   :SetTimer(0 tmrTimer constTimerDelay)

   tgPac.CallActionSlot(calculateNextFrame)

   :LockScreen()
   tgPac.CallActionSlot(updateScreen)
   :UnlockScreen()
 )
}

{:Link lnkKeyText
 :EventSource      0
 :EventType        UserInput
 :EventData        KeyText
 :LinkEffect      ( :Quit() )
}

)

:InputEventReg 4
:SceneCS 720 576
}

```